
Wenn Sie Visual FoxPro einsetzen sind Sie zweifellos auch mit der Funktion CREATEOBJECT() vertraut. Viele Entwickler benutzen diese Funktion direkt und überall innerhalb des Sourcecodes ihrer Anwendung. In vielen Fällen ist dies absolut perfekt, in anderen Fällen ist dieses Vorgehen aber zu unflexibel. 

Damit CREATEOBJECT() korrekt arbeitet müssen Sie einige Bedingungen erfüllen. Zunächst müssen Sie CREATEOBJECT() den Namen der Klassendefinition als Parameter übergeben. Zusätzlich müssen Sie die Programmdatei oder die Klassenbibliothek mit der Klassendefinition mit SET PROC oder SET CLASSLIB geöffnet haben. Wenn das Objekt, das Sie instanziieren wollen, aus Teilen in unterschiedlichen Bibliotheken  oder Programmen zusammengesetzt ist, müssen auch diese geöffnet sein. Sind sie nicht geöffnet, sehen Sie die folgende störende Nachricht: „Klassendefinition <Name> wurde nicht gefunden.“ Ich bin mir sicher, Sie kennen diese Meldung. Es müssen also drei Vorgaben erfüllt sein, bevor Sie ein Objekt anlegen können: der Klassenname (Klassendefinition),  die Klassenbibliothek, die den Klassennamen enthält und eine korrekt eingerichtete Umgebung. Die ganze Geschichte ist also etwas komplexer als es auf den ersten Blick erscheint. Es muss doch einen Weg geben, diese Komplexität zu kapseln und das Erstellen von Objekten in VFP einfacher zu gestalten. Zum Glück gibt es diese Möglichkeit und das Vorgehen, dieses möglich zu machen, ist als Werkstatt bekannt.

In der realen Welt sammeln die Werkstätten Rohstoffe, führen unterschiedliche Arbeitsschritte aus und am Ende erstellen sie ein Produkt. In einer Softwarewerkstatt läuft es ähnlich. Der Rohstoff sind die Dateien (Programme und Klassenbibliotheken), der Prozess ist, wie ich weiter unten ausführen werde, die Klasse und das Produkt ist ein nagelneues VFP-Objekt. Nehmen Sie die einzelnen Teile, setzen Sie sie zusammen und schon haben Sie alle Vorraussetzungen für die Fertigung von Objekten.

Der Rohstoff

Das nächste zu lösende Problem ist die Entscheidung, wo die Rohstoffe gelagert werden sollen. Die einzige logische Lösung ist eine Tabelle. Die Tabelle, ich nenne sie FACTORY.DBF, hat die folgende Struktur:

Jeder Datensatz dieser Tabelle definiert die Information, die erforderlich ist, um ein Objekt zu erstellen. Beachten Sie, dass die Tabelle Felder enthält, um den Namen der Klasse (oClass), die Klassenbibliothek (cLibrary), in der der Name der Klasse definiert wird, zu speichern. Außerdem enthält die Tabelle die Felder mClasses für zusätzliche Klassenbibliotheken und mProcs für Programme, wenn es sich bei dem Objekt um ein Aggregatobjekt handelt. Zusätzlich werden Sie bemerken, dass die FACTORY.DBF noch die Felder cKey, cFile, mPreProc und mPostProc enthält. Diese Felder enthalten sämtliches Rohstoff für die Erstellung eines Objekts.

Die Klassendefinition (cClass)

Dieses Feld enthält den Namen der Klassendefinition, die für die Erstellung des Objekts verwandt werden soll. Der zur Zeit der Erstellung des Objekts in diesem Feld enthaltene Wert wird der Parameter, der der Funktion CREATEOBJECT() übergeben wird. 

Die Klassenbibliothek (cLibrary)

Dieses Feld enthält den Namen der Klassenbibliothek oder Prozedur, in der die Klassendefinition gespeichert ist. Damit wird sichergestellt, dass für die Erstellung des Objekts die richtige Klassenbibliothek oder Prozedur geöffnet wird.

Die unterstützenden Klassen und Prozeduren (mClasses und mProcs)

Beide Felder enthalten eine durch Kommata getrennte Liste von Klassenbibliotheken    oder Prozeduren, die für die Erstellung und Funktionalität des Objekts geöffnet sein müssen. Beachten Sie, dass ein Test durchgeführt werden muss um sicherzustellen, dass eine Klassenbibliothek oder Prozedur nur ein Mal geöffnet ist.

Der Objektschlüssel (cKey)

Bei der Information, die Sie in cKey speichern, handelt es sich um eines der mächtigsten Features der Objektwerkstatt. Dieser Schlüssel ermöglicht es Ihnen, einem Objekt, das Sie erstellen wollen, einen Alias zuzuweisen. Ich nehme jetzt als Beispiel einmal an, dass Sie eine Buchhaltungssoftware erstellen, die es Ihrem Kunden auch ermöglicht, Rechnungen zu erstellen. In dieser Anwendung befindet sich ein Rechnungsobjekt mit der Klassendefinition „boInvoice“, das Ihre Firma zur Verfügung stellt. Wenn Sie nicht mit der Objektwerkstatt oder einem ähnlichen Produkt arbeiten instanziieren Sie Ihre Rechnung mit einem Code, der ungefähr so aussieht:

loInvoice = CREATEOBJ('boInvoice')

Beachten Sie, dass Sie Ihre Anwendung beschränken, wenn Sie den Namen der Klassendefinition im Code hart kodieren. Es ist besser, wenn Sie der Klassendefinition „boInvoice“ einen Alias zuordnen. Ich benutze „InvoiceObject“ als Beispiel für einen Alias. Sie können jetzt das mit einem Alias verknüpfte Objekt mit der Objektwerkstatt einsetzen; die Benutzung einer Wrapperfunktion demonstriere ich Ihnen später. Das Werkstattobjekt sucht die Objektdefinition mit Hilfe des Schlüssels „InvoiceObject“ in der Tabelle FACTORY.DBF, übergibt den Klassennamen der Funktion CREATEOBJECT() und das Produkt ist ein Objekt Ihrer Wahl oder der Ihres Anwenders. 

Wo liegen die Vorteile dieses Vorgehens? Es gibt mehrere Vorteile, der wichtigste ist die Erweiterbarkeit. Nehmen Sie an, Sie haben einen sehr großen Kunden, der mit Ihrem Rechnungsformular nichts anfangen kann. Er benötigt eine erweiterte Version. Wäre es in einem solchen Fall nicht eine gute Verkaufsstrategie, dem Kunden mitzuteilen, dass er von Ihrer Klasse eine weitere Klasse ableiten und dabei all Ihre Funktionalität erben kann, um die Klasse anschließend um die benötigten Funktionalitäten zu erweitern? Anschließend muss Ihr Kunde, um sein neu erstelltes Rechnungsobjekt auch einsetzen zu können, lediglich den Klassennamen und die Klassenbibliothek des Datensatzes „InvoiceObject“ in der Tabelle FACTORY.DBF von „boInvoice“ in „boCustomInvoice“ zu ändern. Nicht schlecht, oder? Beachten Sie dabei auch, dass Ihr ursprünglicher Code in der abgeleiteten Klasse noch vorhanden ist. Damit kann Ihr Kunde auch von Ihren weiteren Updates profitieren, ohne dass seine eigene Arbeit überschrieben wird. 

Die Datei (cFile)

Dieses Feld erweitert den Befehl SET CLASSLIB. Das Format dieses Befehls ist (direkt aus der Hilfedatei von VFP kopiert):

SET CLASSLIB TO ClassLibraryName [IN APPFileName | EXEFileName] [ADDITIVE] [ALIAS Alias]

Beachten Sie, dass sich Klassenbibliotheken auch in anderen Dateien wie Anwendungen, .APP und in .EXE-Dateien befinden können. Wenn sich die Klassenbibliothek, die das zu erstellende Objekt enthält, in einer solchen Datei befindet müssen Sie den Namen dieser Datei angeben. Wenn cFile einen Wert enthält wird der Befehl SET CLASSLIB um die IN-Klausel des Befehls erweitert. Jetzt kann die Funktion CREATEOBJECT() auf eine in einer .APP oder .EXE verborgene Klassenbibliothek zugreifen.

Pre-Processing (mPreProc) und Post-Processing (mPostProc)

Als letzen Schliff habe ich der Objektwerkstatt die zusätzliche Möglichkeit des Pre- und Postprocessing gegeben. Diese Memofelder bieten Ihnen die Möglichkeit, in ihnen ein vollständiges Programm (nicht empfohlen) oder den Aufruf  eines externen Programms (empfohlen) zu platzieren. Diese Programme werden vor und/oder nach der Erstellung des Objekts ausgeführt.

Damit sind alle Teile beschrieben. Ich werde jetzt mit Ihnen durch den Prozess des Zusammenfügens der einzelnen Teile behandeln. Wir erstellen ein Objekt.

Der Prozess

Die Kontrollmethode der Objektwerkstatt ist ein Beispiel für ein verhaltensbedingtes Muster, Template-Methode genannt. Technisch ausgedrückt handelt es sich bei einer Template-Methode um „die Definition des Skeletts eines Algorithmus in einer Operation, wobei einige Schritte in abgeleitete Klassen ausgelagert sind. Template-Methoden ermöglichen es den abgeleiteten Klassen, verschiedene Schritte neu zu definieren, ohne die Struktur des Algorithmus zu ändern“. Der in der Objektwerkstatt eingesetzte Prozess für die Erstellung eines Objekts läuft folgendermaßen ab:

*-- Location Section -------

*} Program: CFactory.PRG

*} Class: CAbstractFactory

*} Method: ManufactureObject()

*---------------------------

LPARAMETERS tcKey, ;

roObject, ;

tlSkipEnvironmentSetup

LOCAL loObject , ;

lcClassName , ;

lcClassLibrary , ;

lcClassFile , ;

lcClassLibList , ;

lcProcList , ;

lcPreProc , ;

lcPostProc , ;

llRetVal

roObject = .NULL.

IF tlSkipEnvironmentSetup

lcClassName = tcKey

ELSE

This.GetClassInformation( tcKey , ;

@lcClassName , ;

@lcClassLibrary , ;

@lcClassFile , ;

@lcClassLibList , ;

@lcProcList , ;

@lcPreProc , ;

@lcPostProc )

ENDIF

IF .NOT. tlSkipEnvironmentSetup

This.SetClassLibraries( lcClassLibrary , ;

lcClassFile , ;

lcClassLibList )

This.SetProcedures( lcProcList )

This.ExecuteProcedure( lcPreProc )

ENDIF

 

roObject = This.InstantiateObject( lcClassName )

IF .NOT. tlSkipEnvironmentSetup

This.ExecuteProcedure( lcPostProc )

ENDIF

 

llRetVal = TYPE('roObject.Name') == "C"

 

RETURN llRetVal

Die Vorteile einer Template-Methode

Ich möchte hier noch einige Anmerkungen zu dem Gebrauch von Template-Methoden machen. Der erste und wichtigste Punkt ist, dass diese Methode eher der natürlichen englischen Sprache entspricht. Es ist klar: Wenn Sie den Schalter tlSkipEnvironmentSetup ignorieren, werden die folgenden Schritte in dieser Reihenfolge ausgeführt, um ein Objekt zu erstellen. Als Erstes müssen Sie sich mit GetClassInformation die Informationen aus der Tabelle, FACTORY holen, anschließend werden SetClassLibraries und SetProcedures aufgerufen, die zur Unterstützung des neu erstellten Objekts erforderlich sind. Nun folgt ExecuteProcedures für das Pre-Processing, gefolgt vom Prozess InstantiateObject. Abschließend wird das Post-Processing, ExecutesProcedures, aufgerufen. Entscheidend ist, dass auch ein unerfahrener Programmierer in weniger als 15 Sekunden erkennt, was hier geschieht. Durch die Template-Methoden wird es für neue Programmierer einfacher, den Code zu verwalten.

Der zweite Punkt: Beachten Sie, wie die Template-Methode es ermöglicht, einzelne Schritte eines Algorithmus neu zu definieren. Nehmen wir an, Sie leiten vom Werkstatt-Objekt eine neue Klasse, DocumentFactory, ab. In dieser neuen Klasse können Sie jeden der Prozesse überschreiben und durch eine anwendungsspezifische Lösung ersetzen. So lange Sie die primäre Methode, ManufactureObject(), nicht überschreiben bleibt die Reihenfolge der Verarbeitung der Methoden erhalten.

Verwenden der Objektwerkstatt, Teil 1 – Das Starterprogramm, Factory.PRG

Bevor ich die Objektwerkstatt im Framework zum ersten mal benutzen konnte, musste ich sie zunächst erstellen. Dies galt natürlich nur für das erste Mal. Ich erstellte das Programm FACTORY.PRG, um die Komplexität des Gebrauchs der Objektwerkstatt zu kapseln. Damit änderte sich der Code, mit dem das Werkstattobjekt angesprochen wird, von 

loInvoice = .NULL.

loFactory = CREATEOBJECT('cAbstractFactory')

IF loFactory.ManfufactureObject('InvoiceObject',;

@loInvoice) = .T.

<do some processing here>

ENDIF

in

loInvoice = .NULL.

IF Factory('InvoiceObject',@loInvoice ) = .T.

<do some processing here>

ENDIF

Diese Methode für den Aufruf der Objektwerkstatt ist einfach aus dem Befehlsfenster heraus aufzurufen. Vorher, innerhalb des Framework, musste ich immer die gesamte Anwendung aufrufen, nur um ein Formular zu testen. Dafür gab es einen einfachen Grund. Das Formular arbeitete nicht korrekt, bevor die Umgebung eingestellt war. Durch die Benutzung der Objektwerkstatt kann ich Formulare aus dem Befehlsfenster heraus testen, ohne erst mühsam die Anwendung starten zu müssen.

Hier das Listing für FACTORY.PRG:

*-- Factory.PRG

LPARAMETER tcKey, roObject, tlSkipEnvironmentSetup

LOCAL lcProcList, llRetVal, loFactory

lcProcList = UPPER(SET('PROC'))

IF .NOT. ( "CFACTORY.FXP" $ lcProcList )

SET PROC TO ..\..\COMMON\LIBS\CFACTORY.FXP ADDITIVE

ENDIF

IF .NOT. ( "IFACTORY.FXP" $ lcProcList )

SET PROC TO ..\..\COMMON\ILIBS\IFACTORY.FXP ADDITIVE

ENDIF

loFactory = CREATEOBJ( 'cAbstractFactory' )

IF TYPE('loFactory.Name') == "C"

loFactory.ManufactureObject( tcKey, ;

@roObject, ;

tlSkipEnvironmentSetup )

llRetVal = TYPE( 'roObject.Name' ) == "C"

ELSE

llRetVal = .F.

ENDIF

RETURN llRetVal

Benutzen der Objektwerkstatt, Teil 2 – Die globale Referenz

Sie haben auch die Möglichkeit, ständig eine Referenz auf die Objektwerkstatt verfügbar zu haben. Im vorigen Beispiel mussten Sie den Prozess zum Erstellen eines Objektes für jedes zu erstellende Objekt zwei Mal durchlaufen. Beim ersten Durchlauf haben Sie die Objektwerkstatt selbst erstellt und beim zweiten Durchlauf wurde durch die Objektwerkstatt eventuell ein Objekt erstellt. Sie könnten jetzt auf den Gedanken kommen, dass dieses Vorgehen nicht effizient ist. In diesem Fall fügen Sie Ihrer globalen Anwendung .oFactory eine Eigenschaft hinzu und während der Instanziierung erstellen Sie die Objektwerkstatt und speichern in der Eigenschaft eine Referenz darauf. Wenn Sie anschließend ein Objekt erstellen wollen müssen Sie lediglich die folgende Signatur benutzen:

<goApp>.oFactory.ManufactureObject( <cKey> , ;

@<oObject> , ;

[<lSkipEnvironmentSetup>] )

Innerhalb einer funktionierenden Anwendung ist dieses Vorgehen in der Regel optimal.

Die Signatur

Wie wird dieses Ding eigentlich benutzt? Als Erstes müssen Sie die FACTORY.DBF mit allen für das Objekt erforderlichen Informationen füllen. Anschließend erstellen Sie eine Variable mit dem neu erstellten Objekt als Inhalt. Zum Abschluss rufen Sie das Werkstatt-Objekt durch den Aufruf seiner Methode ManufactureObject auf. Übergeben Sie dieser Methode als ersten Parameter den mit dem zu erstellenden Objekt verbundenen Schlüssel und als zweiten Parameter die Variable, in der das Objekt gespeichert werden soll. Ein Beispiel:

PRIVATE loInvoice

LoInvoice = .NULL.

IF <goApp>.oFactory.ManufactureObject( "InvoiceObject", ;

@loInvoice )

<work with the object as usual>

ELSE

<perform some exception processing>

ENDIF

Wenn die Methode .T. zurückgibt hat die Objektwerkstatt das Objekt erfolgreich erstellt. Wird .F. zurückgegeben konnte das Objekt aus irgendeinem Grund nicht erstellt werden. 

Unabhängig davon, welche Methode Sie für den Aufruf der Objektwerkstatt verwenden (FACTORY.PRG oder eine andere Objektreferenz) bleibt die Signatur immer gleich. Der erste Parameter ist der in der FACTORY.DBF hinterlegte Schlüssel. Der zweite Parameter ist die Variable, in der das neu erstellte Objekt gespeichert werden soll.

Vereinfachung des Prozesses der Erstellung des Objekts

Die letzte Änderung, die an der Objektwerkstatt vorgenommen werden muss ist die Schaffung der Möglichkeit, sie ohne Auffüllen der FACTORY.DBF einzusetzen. Dafür übergeben Sie der Methode ManufactureObject() einfach ein .T.. Damit werden die Prozesse übersprungen, mit denen die Informationen aus der Tabelle geholt werden. Statt dessen wird der Wert genutzt, der im ersten Parameter des Alias als Klassenname übergeben wurde. Beispielsweise erzeugt der folgende Code ein Objekt aus der Klassendefinition „boInvoice“ unter Umgehung der FACTORY.DBF:

LOCAL loInvoice

loInvoice = .NULL.

<goApp>.oFactory.ManufactureObject("boInvoice, @loInvoice, .T. )

Zusammenfassung

Ich hoffe, dass ich Ihnen den Gebrauch einer Objektwerkstatt für Ihre Objekte, die eine höhere Flexibilität und eine bessere Erweiterbarkeit aufweisen sollen, näher gebracht habe. Folgen Sie grundsätzlich diesen Regeln:

Regel 1: Benutzen Sie die FACTORY.DBF, um Objekte zu erstellen, von denen der Anwender eigene untergeordnete Klassen ableiten kann.

Regel 2: Benutzen Sie die vereinfachte Signatur der Objektwerkstatt, um die Erstellung des Objekts komplett von Ihrer Anwendung zu trennen. Beachten Sie, dass Sie bei der Beachtung dieser Methode nie mehr testen müssen, ob das Objekt erfolgreich erstellt wurde. Sie erfahren dies aus dem Rückgabewert der Methode ManufactureObject().

Je mehr Menschen eine Methodik übernehmen, desto mehr Fehler des ursprünglichen Designs werden innerhalb der verschiedenen Implementierungen deutlich. Sollten Sie in eine solche Situation kommen, in der unser Vorgehen einen Fehler aufweist und geändert werden sollte, setzen Sie sich bitte mit mir in Verbindung und wir werden das Framework neu designen. 

12-10 Eine Ojektwerkstatt
FoxX-Professional
Seite 1

